Representation of multivariate domains

	Topology	2 D	3 D
Domain givem by one implicit equation (=)	Boundary	ContourPlot $D \equiv \{x^2 + y^2 = 1$	ContourPlot3D $D \equiv \{x^2 + y^2 + z^2 = 1$
Domain givem by one or more implicit inequalities (< , ≤)	Interior	RegionPlot $D \equiv \begin{cases} x^2 + y^2 \le 1 \\ x + y > 0 \end{cases}$	RegionPlot3D $D \equiv \{x^2 + y^2 + z^2 \le 1$

Example 1

$D \equiv \{x^2 + y^2 = 1$

As we have two variables we know that it is a domain in $\mathbb{R}^2$ and we use ContourPlot.

```
In[•]:= ContourPlot[x^2 + y^2 == 1, {x, -1, 1}, {y, -1, 1}, Axes → True, Frame → False]
Out[•]=
```



Example 2

$D \equiv \{x^2 + y^2 < 1$

Again we have two variables so D is a domain in the plane defined by inequalities and we use RegionPlot

In[∘]:=
```
RegionPlot[x^2 + y^2 < 1, {x, -1, 1}, {y, -1, 1}, Axes → True, Frame → False]
```
Out[∘]=



Example 3

$$D \equiv \begin{cases} x^2 + y^2 \leq 1 \\ x + y > 0 \end{cases}$$

In[∘]:=
```
graphicinterior = RegionPlot[x^2 + y^2 ≤ 1 && x + y > 0,
  {x, -1, 1}, {y, -1, 1}, Axes → True, Frame → False, PlotPoints → 40]
```
Out[∘]=



We can display in the same graphic two boundary type domains (defined by an identity, =). To representthe domains $D_1 \equiv \{x^2 + y^2 = 1$ and $D_2 \equiv \{x + y = 1$, inside ContourPlot we group both equations with curly braces and separated by comma.

```
In[]:= graphicboundary = ContourPlot[{x^2 + y^2 == 1, x + y == 0},
  {x, -1, 1}, {y, -1, 1}, Axes → True, Frame → False, PlotPoints → 40]
```

Out[]=



As we gave a name to the last two graphics (graphicsinterior, graphicboundary), now we can combine them by means of the instrucion Show.

In[ ]:= `Show[{graphicinterior, graphicboundary}]`

Out[ ]=



Example 4

$D \equiv \left\{ x^3 + y^2 \leq 1 \right.$

In[ ]:= `RegionPlot[x^3 + y^2 ≤ 1, {x, -10, 10}, {y, -20, 20}]`

Out[ ]=



Example 5

$$D \equiv \begin{cases} 3x^2 + 5y^2 \leq 4 \\ 2x + y > 0 \end{cases}$$

In[ ]:=
```
RegionPlot[3 x^2 + 5 y^2 ≤ 4 && 2 x + y ≥ 0, {x, -2, 2}, {y, -2, 2}, Frame → False, Axes → True]
```

Out[ ]=



It is difficult to represent in an exact manner the boundary of this domain since it consists of a part of the boundary domain coming from the first equation , $D_1 \equiv \{3x^2 + 5y^2 = 4$, and a part of the boundary domain defined by the second equation, $D_2 \equiv \{2x + y = 0$. What we can do is to represent in the same graphic both boundary domains, D_1 and D_2.

```
In[=]:= ContourPlot[{3 x^2 + 5 y^2 == 4, 2 x + y == 0}, {x, -2, 2}, {y, -2, 2}, Frame → False, Axes → True]
Out[=]=
```



Example 6

$D \equiv \{x^2 + y^2 + z^2 \leq 1$

In this case we have three variables so D is a domain of the space, $\mathbb{R}^3$, defined by an inequalito so we use RegionPlot3D.

In[ ]:=
```
RegionPlot3D[x^2 + y^2 + z^2 ≤ 1, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```

Out[ ]=



As we can see $D \equiv \{x^2 + y^2 + z^2 \leq 1$ is the closed ball of the space centered at the origin and with radious 1. The boundary of D would be $\partial D \equiv \{x^2 + y^2 + z^2 = 1$ but as it is a closed surface we cannot distinguish the graphic of D and ∂D.

In[ ]:=
```
ContourPlot3D[x^2 + y^2 + z^2 == 1, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```
Out[ ]=



Example 7

$$D \equiv \begin{cases} x^2 + y^2 + z^2 \le 1 \\ x^2 + y^2 \ge z \end{cases}$$

In[ ]:=
```
domainD = RegionPlot3D[x^2 + y^2 + z^2 ≤ 1 && x^2 + y^2 ≥ z,
  {x, -1, 1}, {y, -1, 1}, {z, -1, 1}, PlotPoints → 40]
```
Out[ ]=



As it can be seen, even for Mathematica to represent this domain is a difficult problem and we need to ask for higher precission by means of the option PlotPoints->40.

To have an idea about why the we obtain such representation we have to realize that D is the intersection of $D_1 \equiv \{ x^2 + y^2 + z^2 \le 1$ which is the closed ball or radious 1 centered at the origin and

$D_2 \equiv \left\{ x^2 + y^2 \geq z \right.$. The boundary of D_2 is $\partial D_2 \equiv \left\{ x^2 + y^2 = z \right.$ which is the elliptical paraboloid that can be displayd with ContourPlot3D.

In[∘]:=
```
boundaryD2 = ContourPlot3D[x^2 + y^2 == z, {x, -1, 1}, {y, -1, 1}, {z, -1, 1}]
```

Out[∘]=



D_2 is the domain under this paraboloid and, therefore, D is the part of the ball under such a paraboloid. As we put names to every image we can combine both of them with Show.

In[ ]:=
```
Show[{domainD, boundaryD2}]
```
Out[ ]=



Example 8

$D \equiv \left\{ x^2 e^x + y^2 e^y + z^2 e^z = 1 \right.$

In[ ]:=
```
ContourPlot3D[x^2 e^x + y^2 e^y + z^2 e^z == 1, {x, -15, 1}, {y, -15, 1}, {z, -15, 1}, Axes → True]
```
Out[ ]=



Some important equations:

$\mathbb{R}^2$ (real plane)	
Circunference with radius r centered at p = (x_0, y_0)	$D \equiv \{ (x - x_0)^2 + (y - y_0)^2 = r^2$
Elipse with semi − radius a, b centered at p = (x_0, y_0)	$D \equiv \{ \left(\frac{x-x_0}{a}\right)^2 + \left(\frac{y-y_0}{b}\right)^2 = 1$
Line passing through p = (x_0, y_0) in the direction of v = (a, b)	$D \equiv \{ \left(\frac{x-x_0}{a}\right) - \left(\frac{y-y_0}{b}\right) = 0$
General line with normal vector v = (A, B)	$D \equiv \{ A x + B y = C$
General semiplane	$D \equiv \{ A x + B y \; \begin{smallmatrix} < \\ \leq \\ > \\ \geq \end{smallmatrix} \; C$

$\mathbb{R}^3$ (real space)	
Sphere with radius r centered at p = (x_0, y_0, z_0)	$D \equiv \{ (x - x_0)^2 + (y - y_0)^2 + (z -$
Ellipsoid with semi − radius a, b, c centered at p = (x_0, y_0, z_0) $\mathbb{R}^2$	$D \equiv \{ \left(\frac{x-x_0}{a}\right)^2 + \left(\frac{y-y_0}{b}\right)^2 + \left(\frac{z-y}{c}\right.$
Plane passing through p = (x_0, y_0) with normal vector v = (a, b, c)	$D \equiv \{ a (x - x_0) - b (y - y_0) + c ($
General plane with normal vector v = (A, B, C)	$D \equiv \{ A x + B y + C z = D$
General semi − space	$D \equiv \{ A x + B y + C z \; \begin{smallmatrix} < \\ \leq \\ > \\ \geq \end{smallmatrix} \; I$

Definition of multivariate functions

To define a real function with several variables we use similar notation to the one for one variable. For instance to define the function

$$f(x, y, z) = x^2 + y^2 + \text{Cos}[x\,y\,z]$$

we write in Mathematica:

```
f[x_, y_, z_] := x^2 + y^2 + Cos[x y z]
```

To define a real function with several variables we use similar notation to the one for one variable. For instance to define the function

To define a vectorial multivariate function we have to recall that to represent a vector with Mathematica we use curly braces. Therefore, the vector v=(2,-1,1) is represented in Mathematica by {2,-1,1}

Thus, the function

$$f : \mathbb{R}^2 \longrightarrow \mathbb{R}^3$$
$$f(x, y) = \left(x + y,\ x^2 - y,\ x^2 + y^3\right)$$

we write:

```
f[x_, y_] := {x + y, x^2 - y, x^2 + y^3}
```

Thus, a curve, which is a vectorial curve with one variable is defined in much the same way. For instance, the helix

$$c : [0, 5\pi] \subseteq \mathbb{R} \longrightarrow \mathbb{R}^3$$
$$c(t) = (\cos(t), \sin(t), t)$$

can be defined in Mathematica by

```
In[7]:= c[t_] := {Cos[t], Sin[t], t}
```

Representation of multivariate functions

Representation of a real function of two variables, $f : D \subseteq \mathbb{R}^2 \longrightarrow \mathbb{R}$ as a graph

The only case of multivariate function that can be represented (as a graph of a function) is the one of real functions of two variables, $f : \mathbb{R}^2 \to \mathbb{R}$, that can be dispalyed in the space by meand of the instruction Plot3D

Example 9

$f : [-5, 5] \times [-5, 5] \subseteq \mathbb{R}^2 \to \mathbb{R}$

$f(x, y) = x^2 y + x^2 - y$

```
In[•]:= Plot3D[x^2 y + x^2 - y, {x, -5, 5}, {y, -5, 5}]
```

Out[•]=



In order to display the most “interesting” part of the function Mathematica sometimes truncates de representation in order to keep in the image the main features. In this case we can see that Mathematica truncated the range of values plotted for f(x,y) between -50 and 50. We can ask Mathematica not to apply such a truncation by means of the option PlotRange->All.

In[•]:= `Plot3D[x^2 y + x^2 - y, {x, -5, 5}, {y, -5, 5}, PlotRange → All]`

Out[•]=



Example 9

$f:[-1,1]\times[-1,1]\to\mathbb{R}^2$

$$f(x,y)=\frac{x^2 y}{x^4+y^2}$$

In Plot3D we can insert the complete formula for the function as we did in the example before but if we define the function in advance it is not necessary to copy again the formula as Mathematica will remember the definition.

In[•]:= `f[x_, y_] := x^2 y/(x^4 + y^2)`

In[•]:= `Plot3D[f[x, y], {x, -1, 1}, {y, -1, 1}, PlotPoints → 50]`

Out[•]=



Representation of a curve (vectorial function of one variable, $C : I \subseteq \mathbb{R} \longrightarrow \mathbb{R}^n$)

A vectorial function with one varible can be considered as describing the position of a point in the plane, space, etc. along the time. In other words, we consider the variable of the curve as a parameter representing the time. In that case, the instruction to obtaing the graphic of the path followed by the point is ParametricPlot in the plane and ParametricPlot3D in the space.

Example 10

Represent the curve $\begin{array}{c} c : [0, 5\pi] \subseteq \mathbb{R} \longrightarrow \mathbb{R}^2 \\ c(t) = (t\cos(t),\ t\sin(t)) \end{array}$

In[1]:=
```
ParametricPlot[{t Cos[t], t Sin[t]}, {t, 0, 5 π}]
```

Out[1]=



Example 11

If we defined the curve in Mathematica before, as for any other function, we can use the definition to obtain the graphic without copying the formula again.

```
c[t_] := {Cos[t], Sin[t], t}
```

In[8]:=
```
ParametricPlot3D[c[t], {t, 0, 5 π}]
```

Out[8]=



Example 12

Represent the curve $c:[0, 5\pi] \subseteq \mathbb{R} \longrightarrow \mathbb{R}^2$
$c(t) = (t\cos(t),\ t\sin(t),\ t)$

In[2]:=
```
ParametricPlot3D[{t Cos[t], t Sin[t], t}, {t, 0, 5 π}]
```

Out[2]=